



2

EXCEL IN EXCEL
Excel Advanced

Danny Devriendt

Inhoudsopgave

Inhoud formulebalk in Excel vergroot weergeven.....	1
1 Formules zichtbaar maken	2
2 Inleiding - Probleemoplossend werken in Excel: the max()	2
2-1 De max()-functie	2
3 Opwarming... ..	5
3-1 Waar koop jij jouw TV?	5
3-2 Kip aan 't spit – Pluimveeslachterij De Meersman-Van Driessche	5
4 Gegevens zoeken: vert.zoeken() – horiz.zoeken()	6
4-1 Verticaal zoeken	6
4-2 Horizontaal zoeken	8
5 Handige tools in Excel	9
5-1 Willekeurige waarden – waarden plakken – transponeren	9
5-2 Tekstfuncties en snel aanvullen (flash fill)	9
5-3 Beveiligde werkbladen	10
5-4 Werken met datums	11
5-5 Grootste() – kleinste()	12
5-6 Aantal.als – aantallen.als – som.als – sommen.als	12
5-7 Willekeurige waarden – aselect() – aselecttussen()	13
5-8 Rest() en afronden	14
5-9 Berekening in schijven	15
6 Formulierobjecten in Excel.....	16
6-1 Formulierobjecten toevoegen	16
6-2 De keuzelijst met invoervak	16
6-3 Selectievakjes	19
6-4 Keuzerondjes en groepsvakken	19
6-5 Schuifbalken	21
6-6 Schuifbalken en verschuiven	21
6-7 Oefeningen	24
Aandachtspunten	25
Extra - Flexservice	25
Extra - Weekendtoeslag	25
7 Voorwaardelijke opmaak.....	28
7-1 Gegevens t.o.v. elkaar visualiseren	28
7-2 Voorwaardelijke opmaak verwijderen	29
7-3 Tegenstrijdige voorwaardelijke opmaak	29
7-4 Snel voorwaardelijke opmaak toepassen	30
7-5 Maatwerk	30
7-6 Opmaak op basis van een formule	31
8 Tabellen en bereiken.....	38
8-1 Bereik	38
8-2 Tabellen	39
8-3 Een tabel in een bereik omzetten	42
8-4 Structured references en tabellen in detail	42
9 Draaitabellen	44
9-1 De gegevens	44
9-2 De draaitabel maken	44

9-3	Dynamische databron	46
9-4	Subniveaus weergeven	46
9-5	Filteren	46
9-6	Slicers	47
9-7	Werken met datums	48
9-8	Procentuele voorstelling	49
9-9	Berekende velden	50
9-10	Detailweergave in je draaitabel	51
9-11	Tabblad Analyseren	51
9-12	Tabblad Ontwerpen	51
9-13	Draaigrafieken	52
10	X.Zoeken in Excel	53
10-1	Voorbeeld 1	53
10-2	Voorbeeld 2	54
10-3	Voorbeeld 3	54
10-4	Voorbeeld 4 – geen exacte overeenkomst	55
10-5	Voorbeeld 5 – wildcards	55
10-6	Voorbeeld 6 – geneste X.ZOEKEN	56
10-7	Voorbeeld 7 – Parkeren in de Brugse binnenstad	56
10-8	Voorbeeld 8 – Parkeren in de Brugse binnenstad – uitgebreid	57
10-9	Voorbeeld 9 – Distripost huis-aan-huisbedeling	58
10-10	X.VERGELIJKEN	58
10-11	Voorbeeld 10 – De rijkste Belgen - oefening	58
11	Matrixformules en dynamische arrays	60
11-1	Matrixformules	60
11-2	reeks()	60
11-3	sorteren() en sorteren.op()	61
11-4	uniek()	62
11-5	aselect.matrix()	63
11-6	filter()	63
11-7	Voorbeeld met vert.zoeken(): matrix in een formule	64
11-8	Voorbeeld: dynamische keuzelijst met subkeuzelijst	65
	Inhoudsopgave	67

10 X.Zoeken in Excel

Xzoeken.xlsx

De functie X.ZOEKEN is een krachtige nieuwe functie die voortaan binnen o.a. Excel voor Office 365 ter beschikking is. Helaas is X.ZOEKEN niet in Office 2019 beschikbaar en is er geen achterwaartse compatibiliteit. Goed nadenken over het gebruik van deze nieuwe, krachtige functie dus.

X.ZOEKEN kan zowel gebruikt worden ter vervanging van de bestaande functies VERT.ZOEKEN als van HORIZ.ZOEKEN. Het werkt in beide richtingen en heeft ook enkele belangrijke extra's.

De syntaxis van X.ZOEKEN bestaat uit drie verplichte parameters en drie optionele parameters:

```
X.ZOEKEN(zoekwaarde;zoeken-matrix;matrix_retourneren;[indien_niet_gevonden];  
[overeenkomstmodus];[zoekmodus])
```

10-1 Voorbeeld 1

In het voorbeeld hieronder beschikken we in de kolommen A tot D over informatie van de langste rivieren op aarde. Op basis van een rivier die we in F2 intypen (of kiezen), moet in G2 automatisch de lengte van die rivier verschijnen.

	A	B	C	D	E	F	G
1	River	Country/Continent	Miles	Kilometer		River	Kilometer
2	Nile	Africa	4132	6649,8		Congo	4667,1
3	Amazon	South America	4087	6577,4			
4	Yangtze	Asia	3915	6300,6			
5	Yellow river	Asia	3395	5463,7			
6	Parana	South America	3032	4879,5			
7	Congo	Africa	2900	4667,1			
8	Amur	Asia	2761	4443,4			
9	Lena	Asia	2734	4399,9			
10	Mekong	Asia	2700	4345,2			
11	Mackenzie	Canada	2635	4240,6			
12	Niger	Africa	2600	4184,3			
13	Yenisey	Russia	2543	4092,6			
14	Missouri	United States	2540	4087,7			
15	Mississippi	United States	2340	3765,9			
16	Ob	Russia	2268	3650,0			

Tot nu maken we hier van VERT.ZOEKEN gebruik, zeker als we in F2 een rivier intypen, met volgende formule:

```
=VERT.ZOEKEN($F$2; $A$2:$D$26;4;onwaar)
```

Hetzelfde resultaat bereik je met X.ZOEKEN als volgt:

```
=X.ZOEKEN($F$2;$A$2:$A$26;$D$2:$D$26)
```

Hier merk je al enkele voordelen:

- de kolom waarin je zoekt moet met VERT.ZOEKEN altijd de **eerste** kolom van een bereik zijn. Met X.ZOEKEN speelt dat geen rol meer.
- bij VERT.ZOEKEN moet je de kolommen tellen om te bepalen uit welke kolom je een waarde wilt ophalen. Bij X.ZOEKEN geef je gewoon de kolom in kwestie op als een tweede matrix. Dat betekent dat je ook later niet in de problemen komt, mochten kolommen toegevoegd of verwijderd worden. Met VERT.ZOEKEN moet je eventueel wel aanpassingen doen wat het kolomnummer betreft.
- bij VERT.ZOEKEN moet het benaderen op onwaar ingesteld worden omdat de waarden in de zoekkolom niet alfabetisch gerangschikt zijn. Er moet dus gezocht worden tot er een exacte match is tussen zoekwaarde en gevonden waarde. X.ZOEKEN gaat standaard van een exacte match uit.

Belangrijk is wel dat de kolom waarin gezocht wordt, hetzelfde aantal cellen bevat als de kolom waaruit informatie geretourneerd wordt. Met **Ctrl+Shift+↓** kan je trouwens alle cellen in een kolom selecteren tot juist vóór de eerste lege cel.

Misschien ben je nog niet echt overtuigd van het nut van X.ZOEKEN versus VERT.ZOEKEN, maar met het voorbeeld hieronder lukt dat zeker wel:

10-2 Voorbeeld 2

	A	B	C	D	E	F	G	H	I
1	River	Country/Continent	Miles	Kilometer		River	Country/Continent	Miles	Kilometer
2	Nile	Africa	4132	6649,8		Congo	Africa	2900	4667,1
3	Amazon	South America	4087	6577,4					
4	Yangtze	Asia	3915	6300,6					
5	Yellow river	Asia	3395	5463,7					
6	Parana	South America	3032	4879,5					
7	Congo	Africa	2900	4667,1					
8									

In dit voorbeeld vragen we op basis van de invoer in F2 (naam rivier), tegelijk alle andere informatie op: country/continent, miles en kilometer.

Met VERT.ZOEKEN wordt de formule in G2: =VERT.ZOEKEN(\$F2;\$A\$2:\$D\$26;2;onwaar). Die formule kan je doorvoeren naar de cellen H2 en I2, maar je moet wel manueel het kolomnummer naar respectievelijk 3 en 4 aanpassen.

Met X.ZOEKEN krijg je hier één formule in G2: =X.ZOEKEN(F2;A2:A26;B2:D26). Je zoekt op in kolom A en je retourneert in de derde parameter in één keer 3 kolommen. Resultaat: de formule in G2 loopt automatisch over in H2 en I2 en je krijgt telkens automatisch informatie uit een volgende kolom in je retourbereik: kolom B komt in kolom G; kolom C in kolom H en kolom D in kolom I.

De doorloopformule die automatisch in H2 en I2 verschijnt, kleurt iets lichter.

10-3 Voorbeeld 3

In het voorbeeld hieronder komt in F7 een productcode. Op basis van de tabel in kolommen A en B moet dan in F8 de productnaam verschijnen.

	A	B	C	D	E	F	G
1	Naam	Code					
2	1 kaars H%	108662					
3	2 kaarsen H%	108663					
4	3 en meer kaarsen H%	108664					
5	Abelmoschus esculentus	123217					
6	Abies nordmanniana per bos	430					
7	Abies nordmanniana per kilo	2227					
8	Abies overig per kilo	2221					
9	Abies procera Nobilis Blauw	110707					
10	Abies procera Nobilis Blauw per bos	414					
11	Abies procera Nobilis Blauw per kilo	2226					
12	Abies procera Nobilis Groen per bos	415					

Met VERT.ZOEKEN is dit niet op te lossen, tenzij je de kolommen A en B omwisselt. De opzoekkolom moet bij VERT.ZOEKEN immers altijd de eerste kolom van de tabelmatrix zijn. Bij X.ZOEKEN is dit geen probleem.

Bij X.ZOEKEN voegen we een vierde parameter toe. Die optionele parameter bepaalt wat er gebeurt als de gezochte code niet gevonden wordt. Laat je de vierde parameter leeg, dan verschijnt de foutmelding #N/B als de ingevoerde code niet gevonden wordt of als er (nog) geen code is ingevoerd.

=X.ZOEKEN(F7;B2:B3001;A2:A3001;"Code niet gevonden")

10-4 Voorbeeld 4 – geen exacte overeenkomst

In dit voorbeeld zoeken we de stukprijs op voor de aankoop van jetons, diameter 23.3. Hoe meer stuks, hoe goedkoper de stukprijs. In H4 vul je het gewenste aantal stuks in. In I4 moet dan de stukprijs verschijnen.

	A	B	C	D	E	F	G	H	I
1	Jetons (prijs per stuk)								
2		Type (diameter)							
3	stuks	23.3	25	29	35	rechthoek		#stuks	23.3
4	1000	0,082	0,084	0,074	0,088	0,118		120000	0,042
5	3000	0,073	0,075	0,066	0,077	0,104			
6	6000	0,060	0,062	0,055	0,065	0,091			
7	10000	0,053	0,055	0,050	0,057	0,085			
8	15000	0,051	0,053	0,047	0,055	0,083			
9	25000	0,048	0,051	0,045	0,052	0,078			
10	50000	0,045	0,047	0,042	0,048	0,073			
11	100000	0,042	0,044	0,039	0,045	0,068			
12	200000	0,038	0,040	0,037	0,042	0,064			
13	500000	0,035	0,037	0,036	0,037	0,063			
14									

Hier moeten we dus afwijken van de standaard exacte overeenkomst, want 120000 komt niet in de lijst met stuks voor. 120000 ligt wel tussen 100000 en 200000, dus hier speelt parameter 5, de overeenkomst.

0 - Exacte overeenkomst	1 - Zoekopdracht: eerste-naar-laatste
-1 - Exacte overeenkomst of volgend kleiner item	-1 - Zoekopdracht: laatste-naar-eerste
1 - Exacte overeenkomst of volgend groter item	2 - Binaire zoekopdracht (oplopende volgorde gesorteerd)
2 - Jokertekenoevereenkomst	-2 - Binaire zoekopdracht (aflopende volgorde gesorteerd)

I4: =X.ZOEKEN(H4;A4:A13;B4:B13;;;-1)

Met -1 geef je aan 'exacte overeenkomst of volgend kleiner item'. Als er geen exacte overeenkomst is, wordt naar het kleiner item gekeken. Bij 8000 als zoekwaarde, geldt dus het resultaat voor 6000 en niet 10000. Anders uitgedrukt geldt de eerste kolom als *vanaf*: vanaf 6000 stuks: 0,060, vanaf 10000 stuks 0,053...

10-5 Voorbeeld 5 – wildcards

	A	B	C	D	E	F	G	H	I	J	K	L	M
1	Klantnr	Totaalnaam	Familienaam	Voornaam	Geslacht	Straat	Huisnr	Pnr	Gemeente				
2	KL97102	Bastien Cassiers	Cassiers	Bastien	M	Kloosterweg	99	9700	Oudenaarde			Klant	nuyt
3	KL31273	Lorena Bevernage	Bevernage	Lorena	V	Ernest Claesstraat	2	8560	Wevelgem			Klantnr	KL72237
4	KL54918	Gilles Kerkhove	Kerkhove	Gilles	M	Apollonienstrasse	113	4700	Eupen			Gevonden	Mia Nuytens
5	KL28387	Nell Bonnet	Bonnet	Nell	V	Leliepark	8	9870	Zulte			# match	2
6	KL10465	Ijad Lejoly	Lejoly	iyad	M	Place des Capucins	99	7800	Aat				
7	KL57623	Liya Verhasselt	Verhasselt	Liya	V	Rue de Sardanson	104	5000	Namen				
8	KL75200	Ben Iwens	Iwens	Ben	M	Boskant	13	2260	Westerlo				
9	KL69333	Sem Hanot	Hanot	Sem	M	Kwadeplasstraat	99	1640	Sint-Genesius-Rode				
10	KL87207	Amelia Van Genechten	Van Genechten	Amelia	V	Park	26	7850	Edingen				

In het voorbeeld hierboven, typ je in L2 de naam van een klant in, althans een deel van de naam, bv. *nuyt*. Het resultaat is dat je in L3 het klantnummer krijgt van de eerste klant waarbij *nuyt* ergens in de naam voorkomt. In L4 krijg je dan ook de volledige naam van die eerste klant te zien.

Dat komt omdat:

- in de eerste parameter * toegevoegd wordt vóór en na de ingetypte tekenreeks uit L2. Zo'n ster vertegenwoordigt een willekeurig aantal willekeurige tekens. Een ? vertegenwoordigt één willekeurig teken. Dat zijn *wildcards* of *jokertekens*.

□ in de vijfde parameter een 2 staat. 2 staat voor jokertekenovereenkomst. Daardoor werken * en ? als jokertekens.

L3: =X.ZOEKEN("*" & \$L\$2 & "*" ; \$B\$2:\$B\$757 ; \$A\$2:\$A\$757 ; "no match" ; 2)

L4: =X.ZOEKEN("*" & \$L\$2 & "*" ; \$B\$2:\$B\$757 ; \$B\$2:\$B\$757 ; "no match" ; 2)

In L5 tenslotte tel je het aantal keer dat de tekenreeks uit L2 in een naam voorkomt. Mocht de tekenreeks nergens voorkomen, dan krijg je daar 0. In L3 en L4 krijg je dan *no match*, de tekst die er als vierde parameter van X.ZOEKEN is opgegeven.

L5: =AANTAL.ALS(B2:B757;"*" & L2 & "*")

In het voorbeeld hierboven zijn de *-wildcards in de X.ZOEKEN formule toegevoegd. Het spreekt vanzelf dat die ook weggelaten kunnen worden in de formule en dan mee ingetypt worden bij de zoekstring: *nuyt* met de formule:

L3: =X.ZOEKEN(\$L\$2 ; \$B\$2:\$B\$757 ; \$A\$2:\$A\$757 ; "no match" ; 2)

Het opzoeken is niet hoofdlettergevoelig. Wil je bv. onze bekendste viroloog Marc Van Ranst opzoeken, maar weet je niet of Marc met c of k is en of Van Ranst uit één of twee woorden bestaat, dan is dat de juiste zoekstring: Mar? Van*Ranst, mits instellen van de optie *Jokertekenovereenkomst*.

10-6 Voorbeeld 6 – geneste X.ZOEKEN

	A	B	C	D	E	F	G	H	I	J	
1	Jetons (prijs per stuk)										
2	Type (diameter)										
3	stuks	23.3	25	29	35	rechthoek		#stuks	Type	Prijs	
4	1000	0,082	0,084	0,074	0,088	0,118		8000	29	0,055	
5	3000	0,073	0,075	0,066	0,077	0,104					
6	6000	0,060	0,062	0,055	0,065	0,091					
7	10000	0,053	0,055	0,050	0,057	0,085					
8	15000	0,051	0,053	0,047	0,055	0,083					
9	25000	0,048	0,051	0,045	0,052	0,078					
10	50000	0,045	0,047	0,042	0,048	0,073					
11	100000	0,042	0,044	0,039	0,045	0,068					
12	200000	0,038	0,040	0,037	0,042	0,064					
13	500000	0,035	0,037	0,036	0,037	0,063					
14											

Opnieuw het voorbeeld van de jetons, maar deze keer met twee variabele gegevens: het aantal stuks (H4) en het type (I4). Bestudeer aandachtig de formules die in J4 komen hieronder. Beide formules leveren het juiste resultaat op.

=X.ZOEKEN(H4;A4:A13;X.ZOEKEN(I4;B3:F3;B4:F13);"no match";-1)

=X.ZOEKEN(I4;B3:F3;X.ZOEKEN(H4;A4:A13;B4:F13);"no mach";-1))

10-7 Voorbeeld 7 – Parkeren in de Brugse binnenstad

In dit vereenvoudigd voorbeeld, berekenen we op basis van de tijd van binnen (tijd in) en buiten (tijd uit) rijden in N3 en N4 het te betalen parkeertarief voor de parking Zand in Brugge. Zorg uiteraard voor aangepaste tijd-notatie in de cellen N3 tot N5 en je kunt in N5 eenvoudig de geparkeerde tijd in uur berekenen via aftrekken.

Eerst moeten we een probleem met de tijd oplossen. De tabellen met parkeertarieven zijn immers in decimale uren uitgedrukt.

Achter de tijdweergave in uren en minuten met : als scheiding blijft Excel decimaal rekenen. 1:25 u. is decimaal afgerond 0,059028. Vermenigvuldigd met 24 is dat (afgerond) decimaal 1,42 u.

24 uur is 86400 seconden (24 x 60 x 60). 1 uur is dan ca. 0,041667 (3600 / 86400 seconden) van een dag, zijnde 1/24^e. Vermenigvuldigd de inhoud van de cel N5 met 24 en je krijgt het aantal uren decimaal. 0,041667 * 24 = 1,000008. Eigenlijk 1 want 0,041667 is naar boven afgerond.

	A	B	C	D	E	F	G	H	I	J	K	L	M	N	O	
1		Zonder Pcard					Met Pcard									
2	Uur	Zand	Centrum	Station	Langestraat	Uur	Zand	Centrum	Station	Langestraat	Parking	Zand				
3	1	€ 0,00	€ 1,80	€ 1,10	€ 1,80	1	€ 0,00	€ 1,44	€ 0,88	€ 1,44	Tijd in		9:50			
4	2	€ 3,40	€ 3,60	€ 2,20	€ 3,60	2	€ 1,28	€ 2,88	€ 1,76	€ 2,88	Tijd uit		11:15			
5	3	€ 5,20	€ 5,40	€ 3,30	€ 5,40	3	€ 2,72	€ 4,32	€ 2,64	€ 4,32	Totale tijd		1:25			
6	4	€ 7,00	€ 7,20	€ 4,40	€ 7,20	4	€ 4,16	€ 5,76	€ 3,52	€ 5,76						
7	5	€ 9,00	€ 9,00	€ 5,50	€ 9,00	5	€ 5,60	€ 7,20	€ 4,40	€ 7,20			Standaard	Met Pcard		
8	6	€ 10,80	€ 10,80	€ 5,50	€ 10,80	6	€ 7,20	€ 8,64	€ 4,40	€ 8,64	Te betalen		€ 3,40	€ 1,28		
9	7	€ 12,60	€ 12,60	€ 5,50	€ 12,60	7	€ 8,64	€ 10,08	€ 4,40	€ 10,08						
10	8	€ 14,40	€ 14,40	€ 5,50	€ 14,40	8	€ 10,08	€ 11,52	€ 4,40	€ 11,52						
11	9	€ 14,40	€ 14,40	€ 5,50	€ 14,40	9	€ 11,52	€ 11,52	€ 4,40	€ 11,52						
12																
13																
14																
15																
16																
17																



Dit worden de formules in N8 en O8. Bemerkt dat hier gewerkt wordt met de parameter *exacte overeenkomst of volgend groter item* (1). Immers, elk begonnen uur in de parking geldt als een volledig uur. Wie dus 3 uur en 1 minuut parkeert, betaalt voor 4 uur.

De formule genereert een fout als iemand langer dan 9 uur parkeert. Dit wordt nu voorlopig opgevangen door de foutmelding *max. 9 u.* Hoe kan je dit probleem oplossen?

N8: =X.ZOEKEN(\$N\$5*24;A3:A11;B3:B11;"max. 9 u.";1)

O8: =X.ZOEKEN(\$N\$5*24;G3:G11;H3:H11;"max.9 u.";1)

10-8 Voorbeeld 8 – Parkeren in de Brugse binnenstad – uitgebreid

Bij deze uitbreiding vul je in N2 ook nog de gewenste parking in. Je kunt eventueel ook een keuzelijstje met gegevensvalidatie voorzien. Daarnaast geef je ook nog inrijd- en uitrijdtijd op en bereken je de totale tijd en het te betalen tarief, zowel met als zonder Pcard.

	A	B	C	D	E	F	G	H	I	J	K	L	M	N	O	
1		Zonder Pcard					Met Pcard									
2	Uur	Zand	Centrum	Station	Langestraat	Uur	Zand	Centrum	Station	Langestraat	Parking	Station				
3	1	€ 0,00	€ 1,80	€ 1,10	€ 1,80	1	€ 0,00	€ 1,44	€ 0,88	€ 1,44	Tijd in		9:50			
4	2	€ 3,40	€ 3,60	€ 2,20	€ 3,60	2	€ 1,28	€ 2,88	€ 1,76	€ 2,88	Tijd uit		11:15			
5	3	€ 5,20	€ 5,40	€ 3,30	€ 5,40	3	€ 2,72	€ 4,32	€ 2,64	€ 4,32	Totale tijd		1:25			
6	4	€ 7,00	€ 7,20	€ 4,40	€ 7,20	4	€ 4,16	€ 5,76	€ 3,52	€ 5,76						
7	5	€ 9,00	€ 9,00	€ 5,50	€ 9,00	5	€ 5,60	€ 7,20	€ 4,40	€ 7,20			Standaard	Met Pcard		
8	6	€ 10,80	€ 10,80	€ 5,50	€ 10,80	6	€ 7,20	€ 8,64	€ 4,40	€ 8,64	Te betalen		€ 2,20	€ 1,76		
9	7	€ 12,60	€ 12,60	€ 5,50	€ 12,60	7	€ 8,64	€ 10,08	€ 4,40	€ 10,08						
10	8	€ 14,40	€ 14,40	€ 5,50	€ 14,40	8	€ 10,08	€ 11,52	€ 4,40	€ 11,52						
11	9	€ 14,40	€ 14,40	€ 5,50	€ 14,40	9	€ 11,52	€ 11,52	€ 4,40	€ 11,52						
12																
13																
14																
15																
16																
17																



De formules in N8 en N9:

N8: =X.ZOEKEN(\$N\$5*24;A3:A11;X.ZOEKEN(\$N\$2;B2:E2;B3:E11);"max. 9 u.";1)

N9: =X.ZOEKEN(\$N\$5*24;G3:G11;X.ZOEKEN(\$N\$2;H2:K2;H3:K11);"max. 9 u.";1)

10-9 Voorbeeld 9 – Distripost huis-aan-huisbedeling

Je kunt Distripost inschakelen voor huis-aan-huisbedeling van promotiemateriaal. De prijs wordt door het gewicht van de zending en het aantal te bedelen brievenbussen bepaald.

In het model hieronder vind je per Brugse deelgemeente het aantal brievenbussen. Vul of kies in A18 een deelgemeente en vul in B18 het gewicht van je zending in. Op basis daarvan moet de informatie in de cellen D18, E18 en F18 automatisch berekend worden.

Van mogelijke fouten op mocht in B18 een gewicht hoger dan de max. 130 g ingevoerd worden of vermijd dat die invoer kan.

	A	B	C	D	E	F	G	H	I	J	K	L	M	N	O
1	Brugge	oppervlakte	#inwoners	#huishoudens	#brievenbus	0-19	20-39	40-59	60-79	80+	Distripost huis aan huis bedeling				
2	Assebroek	10,49	19708	8712	9783	4036	4629	5433	4237	1424	Tarieven (max. 130 g)				
3	Binnenstad	4,26	19751	10351	16896	2807	5629	5240	4488	1555	Gewicht		stukprijs		
4	Christus-Koning	1,03	4592	2316	3100	814	1180	1267	1084	352	0		0,093		
5	Dudzele	28,26	2540	1018	1167	640	565	745	490	123	26		0,109		
6	Koolkerke	4,14	3319	1316	1361	769	684	1060	679	127	51		0,135		
7	Lissewege	8,76	2405	1046	1178	549	561	769	467	129	76		0,215		
8	Sint-Andries	19,02	19534	8420	10337	3921	4445	5550	4260	1458	101		0,289		
9	Sint-Jozef	2,46	5302	2396	2639	1037	1178	1599	1226	341					
10	Sint-Kruis	13,93	16225	7107	8975	3116	3626	4389	3882	1203					
11	Sint-Michiels	11,31	12136	5214	6439	2324	2639	3169	2841	1163					
12	Sint-Pieters	15,77	7540	3373	4555	1508	1723	2016	1622	715					
13	Zeebrugge	20,21	4415	774	3842	774	1104	1279	936	251					
14	Zwankendamme	0,84	690	332	302	132	178	235	120	38					
15															
16															
17	Deelgemeente	Gewicht (g)		#Bussen	Stukprijs	Totaal									
18	Sint-Pieters	60		4555	€ 0,14	€ 614,93									
19															

10-10 X.VERGELIJKEN

Naast de functie VERGELIJKEN is nu ook X.VERGELIJKEN als nieuwe functie in Excel beschikbaar. Die functie gaat op zoek naar een overeenkomst en het resultaat is op de hoeveelste positie die overeenkomst er is.

In het voorbeeld hiernaast is het resultaat met ananas (F2) als zoekwaarde in het bereik A3:A36, 7. De 7^e fruitsoort in dat bereik is ananas.

	A	B	C	D	E	F
1	Fruitsoort	Calorische waarde				
2		kJ	kcal		Zoeken	ananas
3	aalbes, rode	159	38	Positie		7
4	aalbes, witte	169	40			
5	aalbes, zwarte	202	48			
6	aardbei	135	32			
7	abrikos, gedroogd	1073	256			
8	abrikos, vers	186	44			
9	ananas	234	56			
10	appel, zonder schil of klokhuis	220	53			
11	avocado	962	230			
12	banaan	341	81			

Dat resultaat kan dan bv. in combinatie met de index() functie verder gebruikt worden.

Vergelijk je X.VERGELIJKEN met VERGELIJKEN, dan kan je bij X.VERGELIJKEN de zoekrichting veranderen. De zoekrichting omdraaien, speelt enkel een rol als de gezochte waarde verschillende keren voorkomt.

In de screenshot hierboven zal X.VERGELIJKEN("aardbei", \$A\$3:\$A\$36;0;1) dezelfde waarde opleveren als X.VERGELIJKEN("aardbei", \$A\$3:\$A\$36;0;-1), namelijk 4.

Mocht *aardbei* evenwel zowel in A6 als in A10 voorkomen, dan zou:

- X.VERGELIJKEN("aardbei", \$A\$3:\$A\$36;0;1) nog altijd 4 opleveren.
- X.VERGELIJKEN("aardbei", \$A\$3:\$A\$36;0;-1) positie 8 opleveren. Er wordt van onderaan gezocht naar een match, maar de positie blijft sowieso altijd van bovenaan geteld.

X.VERGELIJKEN is enkel in Office 365 beschikbaar en werkt zowel verticaal als horizontaal. Je kunt hier ook met jokertekens werken bij het zoeken naar een match.

10-11 Voorbeeld 10 – De rijkste Belgen - oefening

Op basis van de data over de rijkste Belgen, is het de bedoeling dat je info krijgt van zodra je in de cel B3 een naam of een deel van een naam intypt over de ingetypte naam.

Beantwoorden verschillende namen aan de tekenreeks in B3, dan krijg je natuurlijk informatie over de eerste match te zien.

Zorg dat je ook de plaats in de ranking te zien krijgt. De lijst is al op basis van vermogen gesorteerd.

Extra: misschien kan je ook het aantal matches ergens weergeven?

	A	B	C	D	E	F
1		Rijkse Belgen - top 100				
2						
3	Opzoekwaarde	De Nul				
4	Plaats		11			
5	Naam	Familie De Nul				
6	Firma	Jan De Nul				
7	Activiteit	Baggerwerken en bouw				
8	Locatie	Aalst				
9	Vermogen	€ 2 752 250 000,00				
10						
11						
12	Naam	Firma	Activiteit	Locatie	Vermogen	
13	Alexandre Van Damme	AB Inbev	Beheer vermogen	Brussel en Zwitserland	€ 17 295 241 000,00	
14	Familie Werner de Spoelberch	AB Inbev: Stella en Jupiler	Brouwerijen	Vlaams-Brabant	€ 12 524 360 000,00	
15	Familie de Mévius	AB Inbev en Verlinvest	Beheer vermogen	Londen	€ 6 976 250 000,00	
16	Familie Albert Frère	Albert Frère	Beheer vermogen	Charleroi	€ 6 933 410 000,00	
17	Eric Wempeck	M&S&T M&S&T	Beheer vermogen	Milano	€ 3 892 250 000,00	